
Abstract 
This paper proposes a method for the numerical solution of large deflection structural problems 

involving finite strains, subject to exhibiting inelastic constitutive response. This method computes the 

singular point in the arc length method using differential of determinant in TLDL  decomposition. 
Outputs of numerical examples show that the proposed idea works well and the buckling mode may be 
computed from the direct linear equation without additional computational costs. 

 
 
 
１. はじめに 

有限要素法などを使った計算力学の研究開発は，今

や世界の第一線のレベルに到達し，活発な研究開発

が展開されている 1),6),7)．また解析も線形解析だけで

なく，非線形解析を行う必要性が生じますます煩雑

化している．有限要素法の基礎式あるいは境界条件

の非線形性に起因してさまざまな非線形問題が発

生する．例えば，相当応力が降伏点を超え塑性変形

を伴うと，応力・ひずみ関係式は非線形となり，こ

のような非線形問題は材料非線形問題と呼ばれて

いる．また，大たわみ問題(有限変形問題)あるいは

座屈問題(構造安定問題)のようにひずみ・変位関係

として非線形式が仮定される問題を幾何学的非線

形問題と呼んでいる．ひずみ・変形関係の非線形性

に起因して，構造物に不安定な平衡状態の発生を伴

う問題を構造安定問題と呼んでおり，分岐座屈，飛

移り座屈，屈服座屈などがある．これらの構造安定

問題の解析においては，分岐点や極限点近傍におけ

る特異点の数値的取り扱いが問題となるⅠ)-9)． 

 非線形解析には局所線形化による増分法がよく

使われており，荷重増分法や変位増分法および弧長 

法（弧長増分法）などがある．荷重増分法は荷重モ

ードに対する増分量を適切に設定し，逐次的に解い

ていく方法であるが，極限点を超えたつりあい経路

を追跡することは不可能である．変位増分法は適切

な変位増分量に関する右辺に対して荷重パラメー

タの増分量と変位増分箇所以外の変位を求める方

法であり，極限点を超えたつりあい経路を追跡する

ことが可能である．しかし，変位増分量が不明の場

合はこの方法による解析はできない．弧長法は荷重

増分法の欠点をなくすように適切な弧長を設定す

ることにより特異点を含むつりあい経路の追跡に

有効な方法である．いずれの方法もニュートン・ラ

フソン法を基本としており，接線剛性行列の取り方

により修正ニュートン法や準修正ニュートン法お

よびその加速版がある．藤井などは接線剛性行列の

特異点近傍 TLDL 分解時（ L は左下三角行列，D は

対角行列，T は転置）のモードが固有モードに近接

していることを利用して特異点を探索する方法を

提案している 8)．  

柏木は，共役勾配法の理論の中の係数行列の逆行

列算定式とトレース理論を使って固有値問題の行

列式の固有値による微分式を提案した 10)．そこでの

提案式は共役勾配法などの反復法用でありコレス

キー分解などの直接法には使えない．ここでは，共

役勾配法に関する式 10)を参考にして，直接法の代表

的解法である TLDL 分解時に行列式の微分を求めら

れることを示す．また，提案式を弧長法に取り入れ

分岐点や極限点などの特異点を正確に求める方法

を述べている．そこでは，行列式の微分の符号が変

化した時点で，２分法などを使い範囲を狭めながら

特異点を特定できる．また，ニュートン・ラフソン

法における解は行列式の微分によって変化するの

で，それを弧長として利用している．弧長の設定が

自動化できるので，弧長法への影響が大きい基本的

パラメーターを減らすことができる．しかし，提案

式と共にひずみや相当ひずみ量の最大値の絶対値

を制限する必要がある．提案法は密行列の TLDL 分

解や密行列に比べ記憶容量を大幅に節約できるバ

ンド行列の TLDL 分解を対象としている．以下に，

その理論とプログラムおよびアルゴリズムを示す．

また，提案法の妥当性を立体トラス（半谷モデル）

の数値実験を通して検証する． 

 

２．LDLT分解時の行列式の微分 

 固有値問題は次のように表される．A
~
を nn の実

対称行列（有限要素法などでは接線剛性行列と呼ば

れる）とすると，標準固有値問題は 

xx  
~

                            (2.1) 

として示される．ここに，  は固有値を， x は固有

ベクトルを表している．式(2.1)が自明な解をもつた

めには，  
~

が特異でなければならないので 

0]
~

det[                          (2.2) 

ここで，  f を以下のようにおく． 

  ]
~

det[  f                     (2.3) 

トレース理論によって 

 
        



 



 


 11

' ~~~ 



tracetrace
f

f

                                        (2.4) 

 
~

A  























nnn

ij

aa

symaa

a

a





1

2221

11

.
][               (2.5) 

とおく． TLDL 分解時には 
TLDLA                             (2.6) 

になる．ここに， 

 1L を以下のようにおく． 

ILL 1 （ I は単位行列）を展開して整理すると， 

jiiijijiijij gggg
)1()1(3322     (2.10) 

式(2.10)から， ijg は全要素について計算する必要

があるが，バンド幅外の要素については 0ij とな

るので，バンド幅内の ij を使って計算できる．また，

バンド幅が狭いほど計算時間は短くなる． 

 行列式の微分は，式(2.4)より，式(2.6)の逆行列の

対角要素だけを求めて加算すればよい．式

(2.7),(2.8),(2.9)から， 111][  LDLT を展開し対角要

素の和を求めると，式(2.4)は以下のようになる． 

以上から，行列式の微分には修正コレスキー分解

時の D 行列と L 行列の逆行列の要素から成ってい

ることが分かる．上述のように，バンド幅内の要素

を使って計算できるので，密行列の計算を，あたか

もバンド行列の計算をするかのようにできる．よっ

て，バンド行列の計算時間と比べてもそれほどの増

加にならないことが予想される． 

 TLDL 分解プログラムに補足したプログラムによ

り，簡単に ff ' が得られる．この値は大変位弾塑

性有限要素解析などに対し，ニュートン・ラフソン

法による極限点や分岐点の数値解析に便利に利用

できる．従来，この種の探索は固有値問題を解く必

要があり，解析時の負担は膨大であった．しかし，

提案法は連立一次方程式を解く過程で簡単に求め

ることができる方法となっている． 

 提案法のアルゴリズムを示す代わりに Fortran に

よるサブルーチンプログラムを記した．ただし，こ

こでは，数値解析に対しては Fortran が最も適してい

ること，および，注釈行と継続行の修正後，上位バ

ージョンの Fortran にも利用できることから

Fortran77 によるプログラムとした． 

 最初に，密行列に対するプログラムを示す． 

 

C------------------------------------------------------------------------------------ 

C 

C     ** Triangular LDLt decomposition for full matrix 

C 

C     input 

C      A : given symmetric coefficient matrix 

C         2-dimension array as A(NN,N) 

C      B : work vector, 1-dimension array as B(N) 

C      NN : given adjustable dimension for array A 

C      N : given order of matrix A and vector B 

C      EPS : parameter to check singularity of the matrix output 

C     output 

C      A : L matrix and D matrix, 2-dimension array as A(NN,N) 

C      FD : differential of determinant 

C      ICHG : numbers of minus element of diagonal matrix D 

C             (numbers of eigenvalue) 

C      IERR : error code 

C             =0, for normal execution 

C             =1, for singularity 

C 

      SUBROUTINE LDLV(NN,N,A,B,FD,EPS,ICHG,IERR) 

C 

      IMPLICIT REAL*8 (A-H,O-Z) 

      DIMENSION A(NN,N),B(N) 

C 

C     LDLt decomposition 

      ICHG=0 

      DO I=1,N 

C       d(i,i) 

        DO K=1,I-1 

          A(I,I)=A(I,I)-A(K,K)*A(I,K)**2 

        END DO 

        IF (A(I,I).LT.0.D0) ICHG=ICHG+1 

        IF (DABS(A(I,I)).LT.EPS) THEN 

          IERR=1 

          RETURN 

        END IF 

C       l(i,j) 

        DO J=I+1,N 

          DO K=1,I-1 

            A(J,I)=A(J,I)-A(J,K)*A(K,K)*A(I,K) 

          END DO 

          A(J,I)=A(J,I)/A(I,I) 

        END DO 

      END DO 

      IERR=0 

C 

      FD=0.D0 

      DO I=1,N 

C       (i,i) 

        FD=FD-1.D0/A(I,I) 

C       (i,j) 

        DO J=I+1,N 

          B(J)=-A(J,I) 

          DO K=1,J-I-1 

            B(J)=B(J)-A(J,I+K)*B(I+K) 

          END DO 

          FD=FD-B(J)*B(J)/A(J,J) 

        END DO 

      END DO 

C 

      RETURN 

      END 

C------------------------------------------------------------------------------------ 

C 

C    ** Solution of linear equation by LDLt for full matrix 

C 

C    input 

C     A : L matrix and D matrix, 2-dimension array as A(NN,N) 

C     B : given right hand side vector, 1-dimension array as B(N) 

C     NN : given adjustable dimension for array A 

C     N : given order of matrix A and vector B 

C    output 

C     B : work and solution vector, 1-dimension array as B(N) 

C 

      SUBROUTINE SOLV(NN,N,A,B) 

C 

      IMPLICIT REAL*8 (A-H,O-Z) 

      DIMENSION A(NN,N),B(N) 

C 

C     forward substitution 

      DO I=1,N 

        DO J=I+1,N 

          B(J)=B(J)-A(J,I)*B(I) 

        END DO 

      END DO 

C 

C     backward substitution 

      DO I=1,N 

        B(I)=B(I)/A(I,I) 

      END DO 

      DO I=1,N 

        II=N-I+1 

        DO J=1,II-1 

          B(J)=B(J)-A(II,J)*B(II) 

        END DO 

      END DO 

C 

      RETURN 

      END 

C------------------------------------------------------------------------------------ 

 

 次に，バンド行列に対するプログラムを示す．以

下にバンド行列の概要を図-1 に示す．ここに，nq は

対角要素を含むバンド幅を示している． 

 

 

 

 

 

 

 

 

図-1 バンド行列 

 

C------------------------------------------------------------------------------------ 

C 

C    ** Triangular LDLt decomposition for band matrix 

C 

C    input 

C     A : given coefficient band matrix  

C        2-dimension array as A(NN,NQ) 

C     B : work vector, 1-dimension array as B(N) 

C     NN : given adjustable dimension for array A 

C     N : given order of matrix A 

C     NQ : given half band width of matrix A 

C     EPS : parameter to check singularity of the matrix 

C    output 

C     A : L matrix and D matrix, 2-dimension array as A(NN,NQ) 

C     FD : differential of determinant 

C     ICHG : numbers of minus element of diagonal matrix D 

C            (numbers of eigenvalue) 

C     IERR : error code 

C            =0, for normal execution 

C            =1, for singularity 

C 

      SUBROUTINE MCDHB(NN,N,NQ, A,B,FD,S,ICHG,IERR) 

C 

      IMPLICIT REAL*8 (A-H,O-Z) 

      DIMENSION A(NN,NQ),B(N) 

C 

C     LDLt decomposition 

      ICHG=0 

      DO I=1,N 

C       d(i,i) 

        DO J=MAX(1,I-NQ+1),I-1 

           A(I,NQ)=A(I,NQ) -A(J,NQ)*A(I,NQ+J-I)**2 

        END DO 

        IF (A(I,NQ).LT.0.D0) ICHG=ICHG+1 

        IF (DABS(A(I,NQ)).LT.EPS) THEN 

          IERR=1 

          RETURN 

        END IF 

C       l(i,j)  : i>j 

C       vertical decomp. 

        DO J=I+1,MIN(I+NQ-1,N) 

          AA=A(J,NQ+I-J) 

          DO K=MAX(1,J-NQ+1),I-1 

            AA=AA- A(I,NQ+K-I)*A(K,NQ)*A(J,NQ+K-J) 

          END DO 

          A(J,NQ+I-J)=AA/A(I,NQ) 

        END DO 

C 

      END DO 

      IERR=0 

C 

      FD=0.D0 

      DO I=1,N 

C       (i,i) 

        FD=FD-1.D0/A(I,NQ) 

C       (i,j) 

        DO J=I+1,MIN(I+NQ-1,N) 

          B(J)=-A(J,NQ-(J-I)) 

          DO K=1,J-I-1 

            B(J)=B(J)-A(J,NQ-(J-I)+K)*B(I+K) 

          END DO 

          FD=FD-B(J)*B(J)/A(J,NQ) 

        END DO 

        DO J=I+NQ,N 

 

nq  

n  n

nq  

          B(J)=0.D0 

          DO K=1,NQ-1 

            B(J)=B(J)-A(J,K)*B(J-NQ+K) 

          END DO 

          FD=FD-B(J)*B(J)/A(J,NQ) 

        END DO 

      END DO 

C 

      RETURN 

      END 

C------------------------------------------------------------------------------------ 

C    ** Solution of linear equation by LDLt for band matrix 

C 

C    input 

C     A : given decomposed coefficient band matrix 

C        2-dimension array as A(NN,NQ) 

C     B : given right hand side vector, 1-dimension array as B(N) 

C     NN : given adjustable dimension for array A 

C     N : given order of matrix A and vector B 

C     NQ : given half band width of matrix A 

C    output 

C     B : solution vector, 1-dimension array as B(N) 

C 

      SUBROUTINE SOLHB(NN,N,NQ,A,B) 

C 

      IMPLICIT REAL*8 (A-H,O-Z) 

      DIMENSION A(NN,NQ),B(N) 

C 

C     forward substitution 

      DO I=1,N 

        DO J=MAX(1,I-NQ+1),I-1 

          B(I)=B(I)-A(I,NQ+J-I)*B(J) 

        END DO 

      END DO 

C 

C     backward substitution 

      DO I=1,N 

        II=N-I+1 

        B(II)=B(II)/A(II,NQ) 

        DO J=II+1,MIN(N,II+NQ-1) 

          B(II)=B(II)-A(J,NQ+II-J)*B(J) 

        END DO 

      END DO 

C 

      RETURN 

      END 

C------------------------------------------------------------------------------------ 

 

３．提案法による非線形 FEM のアルゴリズム 

 非線形有限要素法における特異点の行列式は 0 で

あり，固有値も 0 である．行列式の固有値に関する

微分式があれば，特異点の近傍では微分の符号が変

化するので特異点の探索は容易となる．弧長法をベ

ースとして， ff /' の符号が変化した時点で，２分

法などを使い範囲を狭めながら特異点を特定でき

る．もちろん TLDL 分解時の対角行列 D の負の個数

によっても計算できる．また，ニュートン・ラフソ

ン法における解は )/'/(1 ff によって変化するので，

|)/'/(1| ff を弧長として利用できる．弧長の設定が

自動化できるので，弧長法への影響が大きい基本的

パラメーターを減らすことができる．しかし，筆者

のこれまでの数値実験から，分岐点や極限点などの

特異点を正確に求めるには， |)/'/(1| ff と共にひず

みや相当ひずみ量の最大値の絶対値を制限する必

要がある．例えば，鋼材を使う場合は，降伏ひずみ

をある数値で除した値（ひずみ制限値）で制限する

ことになる．よって， |)/'/(1| ff とひずみ制限値の

小さい方を利用すれば正確に特異点を探索できる．

特に上述の事項を考慮すると以下のような主経路

の途中で特異点を求める弧長法のアルゴリズムを

得る． 

(1) step=0 

 ①データ設定あるいは入力（増分時収束率，最大

ステップ数，１ステップでの最大反復回数，弾

性解析と弾塑性解析の区別，ひずみ制限値用分

割個数， TLDL 分解時特異性判定値，弾性係数

や塑性に関するパラメーター，節点座標，各要

素特性，境界条件など） 

 ②バンド幅の計算 

 ③配列などの初期化 

 ④外力ベクトルの設定あるいは入力 

(2) step=1,2,3,… 

 1) iteration=0 

  ⑤前ステップの情報（節点座標，断面性能，変

位，ひずみ度，応力， ff /' ，…）を記憶 

  ⑥接線剛性行列の計算 

2) iteration=1,2,3,… 

  ⑦ TLDL 分解（ ff /' と対角行列 D の負の個数

計算を含む）および連立 1 次方程式の解算出 

  ⑧弧長法による値と |/'|/1 ff の絶対値の小さ

い方を弧長とする．反復時ひずみ度の最大値

がひずみ制限値を超えないように調整．増分

ひずみと全ひずみを計算 

  ⑨構成則により任意点での材料剛性や応力な

どを計算 

⑩接線剛性行列の計算 

  ⑪残差（不つりあい量）の計算と収束チェック 

  ⑫収束していなければ 2)へ．収束していれば，

ff /' の符号の変化があった場合，2 分法に

より特異点を探す．十分な精度の特異点が得

られたら， TLDL 分解時の対角行列 D の負の

個数により確認し，次ステップ(2)へ進む．た

だし，最大ステップ数に達した場合，計算を

ストップする 

 

４．数値実験 
TLDL 分解時の行列式の微分の弧長法への有効性

を示すために，立体トラス（半谷モデル 4),5),8), 

図-2）による数値実験を行った．図-2 から分かるよ

うに，この立体トラスは 24 部材から成り 13 節点を

構成している対称架構（X-Y 平面）である．また，

荷重は対称変形を成すように 1～7 節点に鉛直下向

きに作用させ，１節点はそれ以外の節点荷重の 2 倍

とした．ここでの数値実験は，すべて倍精度演算と

し，前節のアルゴリズムに従った．計算に

は ， ’Intel(R) Core™ i7 3.2GHz, RAM12.0 GB,  

Windows 7, gcc-4.7-20110723-64 Fortran’を使用した．

また，密行列（式(2.4)，(2.11)）とバンド行列（式(2.11)）

の３方法により解析し，同一結果を得た．提案式に

より ff /' を正確に求め得ることを確認した．以降

では，バンド行列による解析結果を示す． 

以下にデータの設定値を記す．増分時収束率は
610TOLER ，最大ステップ数は 2000NSTEP ，

１ステップでの最大反復回数は 20NITR ，弾性解

析（ 0IEP ）と弾塑性解析（ 1IEP ）の両方，ひ

ずみ制限値用分割個数MD は 30,40,50,60の 4通りと

した． 
 
 
 
 
 
 
 

 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 

図-2 立体トラス（半谷モデル） 

 
TLDL 分解時特異性判定値 1210EPS ，弾性係数や

塑 性 に 関 す る パ ラ メ ー タ ー は 弾 性 係 数 
510058.2 E (N/mm2)，初期断面積 0.10 A (mm2)，

ポアソン比 3.0 (弾性範囲), 5.0 (塑性範囲)，降

伏点 2.235y (N/mm2)．ここでは，応力をひずみ度

から陽に計算できるように，構成則は Richard-Abbott

の構成則 9)とした．応力度 とひずみ度  の関係を

式(4-1),(4.2)に示す． 

ここに， pE は見かけ上のひずみ硬化率を示し，本 

図-3 鉛直変位（節点 1）と固有値など（MD=30） 

図-4 鉛直変位（節点 1）と固有値など（MD=40） 

図-5 鉛直変位（節点 1）と固有値など（MD=50） 

図-6 鉛直変位（節点 1）と固有値など（MD=60） 
 

数値実験では EEp 01.0 とした．mは滑らかさの度

合いを示すパラメーターで 18m とした．また， tE

は任意の点での接線剛性である． 

参考文献 2)を基にして立体トラスの接線剛性マ

トリクスを誘導すると以下のようになる． 

   N は軸力， i と j は部材両端の節点番号， 

),,( iii zyx と ),,( jjj zyx は ji, 節点の zyx ,,  

座標， は部材長である． 

 図-5,6 は固有値の正しい個数を描いている．極限

点までに 6 個の固有値が存在し 5)，6 番目の固有値

が極限点のものである．図-3～図-6はMD毎のRAM-

鉛直変位グラフを描いており，MD の変化による固

有値の個数の変化を描いている． 1 から 5 は分岐

座屈点を， 6 は極限点を表している． 2 と 3 およ

び 4 と 5 はそれぞれに重根となっている．図-3～図

-6で分かるように，全体的にはほぼ同じ経路を辿り，

固有値の個数や分布に差は殆どない．しかし，MD

の値が小さい場合、提案法によって求めた 0 固有値

の個数に差があり（図-3 では 3 個，図-4 では 4 個，

図-5,6 では 6 個），特異点を見落としている．MD の

個数が固有値の正確度に影響を及ぼすことがわか

る．ここでの数値実験では，MD は 50 以上で正しい

解を得ている．このように精度の良い非線形解析は

数多くのステップ数を要することを示している．提

案法は連立一次方程式を解く過程で簡単に求める

ことができる方法であり，非線形解析に便利に利用

されうる方法となっている． 
 
４．結び 

直接法の代表的解法である TLDL 分解時に同時に

行列式の微分を求められることを示した．また，提

案式を弧長法に取り入れ分岐点や極限点などの特

異点を正確に求める方法を述べた．そこでは，行列

式の微分の符号が変化した時点で，２分法などを使

い範囲を狭めながら特異点を特定できる．また，ニ

ュートン・ラフソン法における解は行列式の微分に

よって変化するので，それを弧長として利用するこ

とが可能である．弧長の設定が自動化できるので，

弧長法への影響が大きい基本的パラメーターを減

らすことができる．しかし，数値実験の結果からも

分かるように，提案式と共にひずみや相当ひずみ量

の最大値の絶対値を制限する必要があり，ひずみ制

限値で増分量を制御する必要がある．提案法は
TLDL 分解を対象としており，バンド行列は密行列

の TLDL 分解に比べ記憶容量を大幅に節約できる． 

立体トラス骨組の数値実験結果から，提案の剛性

行列の固有方程式に対する特性方程式の微分値は

特異点（分岐座屈点や極限点）を的確に探索できる．

この方法は多くの時間を必要とする固有値問題を

解く必要が無く，増分解析時の連立 1 次方程式の解

析結果を利用できるので，非常に有効な方法と思わ

れる．今後，多くの数値実験により本法の有効性を

確認する予定である． 

